
Част 2. Microsoft Excel 

2.1.	Формат на състезанието в секция „Microsoft Excel“
Състезанието по информатика в секция „Microsoft Excel“ цели да провери 

възможностите за решаване на практически задачи със средствата на Microsoft 
Excel. Участниците получават файл с въведени данни, оформени в таблични об-
ласти. Заданието за състезанието включва няколко типа задачи:

1)	Въвеждане на подходящи формули.
2)	Обобщаване на данни от таблица с използване на функции или обобщава-

щи таблици (Pivot tables).
3)	Създаване и оформяне на графики (Chart и Pivot Chart).
4)	Форматиране на таблици, в това число условно форматиране.
5)	Създаване и прилагане на ограничения върху въвежданите данни (Data 

Validation).
За решаване на задачите участниците трябва да могат да създават формули 

с използване на: аритметични оператори; абсолютни, относителни и смесени ад-
реси; имена на клетки и области от клетки и вградени функции. Изисква се добро 
познаване на най-често използваните функции в Excel в различни категории:

•• Math&Trig – ABS, MOD, RANDBETWEEN, ROUND, ROUNDDOWN, 
ROUNDUP, SUM, SUMIF, SUMIFS, SUMPRODUCT и др.

•• Statistical – AVERAGE, AVERAGEIF, AVERAGEIFS, COUNT, COUNTIF, 
COUNTIFS, LARGE, MIN, MAX, SMALL и др.

•• Lookup & Reference – HLOOKUP, VLOOKUP, MATCH, INDEX, CHOOSE 
и др.

•• Logical – IF, IFERROR, IFNA, AND, OR.
•• Text – TEXT, LEFT, MID, RIGHT, LEN, CONCATENATE и др.
•• Date & Time – NOW, TODAY, DATE, DAY, MONTH, WEEKDAY, 

WEEKNUM, TIME, YEAR, YEARFRAC.
•• Information – ISBLANK, ISERR, ISERROR и др.

Списъкът с функции не е изчерпателен. Състезателите трябва да могат да 
се ориентират в различните категории и да боравят с помощта на Excel, ако се 
налага, да използват нови за тях функции.

Библиотеката на Excel 2016 съдържа 471 функции, като с всяка нова версия 
техният брой непрекъснато се разширява. Това ръководство няма за цел да пред-
стави всички функции, а се опитва да обърне внимание на най-често използвани-
те при решаване на задачите от състезанието. Тъй като е предназначено за учени-
ци, които имат много добра и отлична подготовка за работа с Excel от часовете по 
информационни технологии в училище, не се разглеждат най-разпространените 

функции като SUM, MIN, MAX, AVERAGE, COUNT и др., защото се предполага, 
че състезателите са запознати добре с тях.

Предвид ограничения обем на ръководството няма как да се обхванат всички 
възможни типове задачи и функции, които може да се изискват при решаване на 
условията. Акцентът е върху задачи, които са част от заданието за всяко състе-
зание в тази секця – като обобщаване на данни и свързване на таблици. Наред с 
това се представят и редица вградени функции за работа с текстови низове, дати, 
логически типове данни и др. Разглеждат се различни инструменти за решаване 
на тези задачи, както и ограничения върху въвежданите данни, условно форма-
тиране и др. 

В края на втората част от ръководството се разясняват две комплексни зада-
ния, дадено е и примерно задание за самоподготовка.

На уеб сайта на състезанието по информатика (http://informatics.ue-varna.bg/
contest.html) са качени всички файлове, използвани в ръководството. За по-добра 
подготовка се препоръчва работа едновременно с ръководството и файловете, 
както и тестване на предложените начини за решаване на задачите. Следва да се 
има предвид, че в някои случаи е възможен повече от един вариант на решение, 
като невинаги всички варианти са представени в ръководството.

2.2.	Обобщаване на данни в Excel
Обобщаването на данни представлява изчисляване на агрегирани стойности 

от таблици с данни. Агрегирането може да бъде по различни критерии (направ-
ления) и с използване на различни функции като SUM, MIN, MAX, AVERAGE, 
COUNT и др.

В Excel има два основни начина за обобщаване на данни – чрез вградени 
функции и чрез обобщаващи (пивот) таблици.

2.2.1.	 Функции за обобщаване по критерии

Към тази група функции спадат SUMIF, SUMIFS, COUNTIF, COUNTIFS, 
AVERAGEIF, AVERAGEIFS. Тези функции се използват за обобщаване (сумира-
не, броене, усредняване) на стойности, които отговарят на определени условия 
или на които съответстват клетки, отговарящи на определени условия. Функци-
ите, завършващи на „IF“, позволяват да се зададе само едно условие с участието 
на операторите за сравнение: „>“, „<“, „>=“, „<=“,“<>“, а тези, завършващи на 
„IFS“ – повече от едно условия.

Общият формат на функция за сумиране по едно условие е:
(f1)8     =SUMIF (range, criteria [, sum_range])

8	 Формулите в помагалото са номерирани, за да е по-ясно позоваването на тях в текста.

Функцията SUMIF проверява всяка от клетките в range дали отговаря на 
условието, зададено в criteria. За клетките, отговарящи на условието, се сумират 
стойностите на съответстващите им клетки в sum_range. Последният аргумент 
sum_range е незадължителен, т.е. ако не се зададе, се приема, че проверяваните 
клетки в аргумента range са и клетките, които ще се сумират. Подобен е принци-
път на работа на функциите COUNTIF и AVERAGEIF.

Общият формат на функция за сумиране по няколко условия е:
(f2)	 =SUMIFS (sum_range, criteria_range1, criteria1, criteria_range2, 

criteria2,...)
Функцията SUMIFS сумира стойностите на клетките в sum_range, за които 

съответстващите им клетки в criteria_range1,  criteria_range2 и т.н. отговарят на 
зададените критерии – criteria1, criteria2 и т.н. 

Приложението на функциите за условно обобщение може да се илюстрира с 
примерна задача. На фиг. 2.1 е показана таблица с месечни продажби за октомври 
2018 г. в три магазина на една компания. В първата колона се въвеждат датите 
(магазините нямат почивен ден), в колона „Магазин“ се посочва градът, в който 
се намира търговският обект (Варна, Русе или София), в колона „Стойност“ са 
отразени продажбите за съответната дата, а в колона „Ден“ с функция се извежда 
денят от седмицата. Възможно е да се използват различни вградени функции за 
извеждане на деня от седмицата, например:

(f3)	  =TEXT(A4;“dddd”)
(f4) 	 =CHOOSE(WEEKDAY(A4;2);“понеделник“;“вторник“;“сряда“; 

“четвъртък“;“петък“;“събота“;“неделя“)
Може да се създаде и помощна таблица с имената на дните от седмицата 

и да се използват функции от групата Lookup & Reference като VLOOKUP или 
HLOOKUP (повече за функциите от тази група ще намерите в т. 2.3.2). Вариантът 
с използване на функция TEXT (f3) е значително по-опростен от този с функция 
CHOOSE (f4) и е за предпочитане. Въпреки това при решаване на задача за състе-
зание ще се признават всички верни решения.



32	 Ръководство за подготовка за национално състезание по информатика

Фиг. 2.1. Месечни продажби по дати и магазини

За улеснение при въвеждането на формулите са дефинирани имена на обла-
стите от клетки в колоните в таблицата. Използването на имена на клетки вместо 
техните адреси е препоръчителна практика, когато едни и същи адреси участват 
в множество формули и когато искаме да подобрим читаемостта на формулите. 
Един лесен и бърз начин за дефиниране на имена на клетки, които са оформени 
в таблица, е с използване на инструмента  след маркиране 
на всички клетки от таблицата. В примера от фиг. 2.1 данните са организирани 
по колони, затова при дефиниране на имената се маркира опцията „Top Row“. 
Разбира се, може да се дефинират имена и по обичайния начин – с инструмента 

. 
Въз основа на данните за продажбите мениджърите на търговската верига 

могат да формулират множество въпроси, свързани с продажбите. Например: 
„Каква е общата сума на продажбите в гр. Варна?“, „Колко са среднодневните 
продажби в понеделник за гр. София?“, „Колко дни в гр. Русе е имало продажби 
над средните за периода?“, „Какво е съотношението между среднодневните про-
дажби през работната седмица и през уикенда?“ и т.н. На тези и подобни въпроси 
може да се намери отговор с помощта на функциите SUMIF(S), COUNTIF(S) и 
AVERAGEIF(S). В таблица 2.1 са представени няколко примера за обобщаване на 
данните с тези функции. Файлът с данните може да се изтегли от сайта на със-
тезанието (http://informatics.ue-varna.bg/contest.html) и да се тестват показаните в 
таблицата формули. Биха могли да се измислят и други подобни примери.

Критериите, използвани във функциите от тази група, се задават като тек-

стов низ или адрес. В таблица 2.1 всички примери са с използване на текстов низ. 
В пример (2) се използва оператор за сравнение „>=“. В този случай операторът 
се огражда с кавички и се използва операторът „&“ за слепване на текстови низо-
ве. В Excel датите имат числова стойност, т.нар. „серийно число“, което е равно 
на броя дни от 1.01.1900 г. до съответната дата. Функцията DATE във формулата 
връща серийното число на дата по зададени година, месец и ден. Друг вариант за 
дефиниране на критерия в пример (2) е да се въведе в клетка датата 15-10-2018 г. 
и след това да се посочи адресът на клетката вместо функцията DATE. 

Таблица 2.1
Примери за използване на SUMIF(S), COUNTIF(S), AVERAGEIF(S)

Пример Формула
1 Продажби общо за Варна  =SUMIF(Магазин;"Варна"; Стойност) 
2 Продажби във Варна след 

15.10.2018 г.
 =SUMIFS(Стойност;Магазин; "Варна";Дата;">="& 
DATE(2018;10;15)) 

3 Продажби в София за периода  
5-15.10.2018 г.

 =SUMIFS(Стойност;Магазин; "София";Дата;">=" & 
DATE(2018;10;5);Дата;"<=" & DATE(2018;10;15)) 

4 Продажби за Варна в понедел-
ник и вторник

=SUM(SUMIFS(Стойност;Ден; {"понеделник","вторник
"};Магазин;"Варна"))

5 Брой дни в Русе с продажби 
над средните за периода

=COUNTIFS(Магазин;"Русе";Стойност;">="&AVERAG
E(Стойност))

6 Брой дни в Русе с продажби 
над средните за града

=COUNTIFS(Магазин;"Русе";Стойност;">="&AVERAG
EIF(Магазин;"Русе";Стойност))

7 Среднодневни продажби в по-
неделник за София

=AVERAGEIFS(Стойност;Магазин;"София";Ден;"понед
елник")

В пример (3) се използват два критерия за датата – тя трябва да е в периода 
от 5.10.2018 г. до 15.10.2018 г. Функциите SUMIFS, COUNTIFS и AVERAGEIFS 
обединяват всички критерии с логически оператор AND, т.е. обобщават данни от 
клетки, които отговарят едновременно на всички критерии, зададени като двойка 
аргументи (criteria_range, criteria). 

В пример 4 е показано използването на масив от стойности – {„понедел-
ник“, „вторник“}. Масивите от стойности задължително се ограждат с „къдра-
ви“ скоби. Тъй като масивът съдържа две стойности, то и SUMIFS ще върне два 
резултата – един за стойността на продажбите във Варна в понеделник и друг за 
продажбите във Варна във вторник. Необходимо е да се сумират тези резултати, 
затова SUMIFS се задава като аргумент на функция SUM. Показаната в пример 
(4) формула е равносилна на следната (f5):

(f5) =SUMIFS (Стойност; Ден; “понеделник“; Магазин;“Варна“)+ 
SUMIFS(Стойност; Ден;“вторник“;Магазин;“Варна“).

В пример (5) и пример (6) се използва функцията AVERAGEIFS. Разли-
ката между тях е, че в пример (5) се извежда броят на всички дни, в които про-
дажбите в Русе са били повече от средните за целия период за всички магазини 
(AVERAGE(Стойност) = 3244,44 лв.), докато в пример (6) се изчислява броят на 
всички дни, в които продажбите в Русе са били повече от средната стойност на 
продажбите в града (AVERAGEIF (магазин; Русе; стойност) = 3114,87 лв.). 

За да се изчисли средната стойност на продажбите от работните дни и през 
уикенда, може да се да използва функция AVERAGEIF с критерий масив, но след-
ва да се има предвид, че получената стойност няма да бъде средноаритметична, 
а среднопретеглена. Така например формулата за изчисляване на среднодневните 
продажби през работната седмица с използване на критерий масив ще бъде тази: 

(f6) =AVERAGE(AVERAGEIF(Ден;{“понеделник“;“вторник“;“сряда“; 
“четвъртък“;“петък“};стойност)) ,
но нейният резултат (3190,75 лв.) няма да е равен на средната стойност на всички 
продажби през работната седмица (3187,29  лв.). Разликата се получава от раз-
личния брой понеделници, вторници и т.н. в рамките на месеца. Ако се търси 
действителната средна стойност на дневните продажби през работната седмица, 
трябва да се добави допълнителна колона „Уикенд/работна седмица“ към табли-
цата от фиг. 2.1. В тази колона, с помощта на подходящи функции, ще се извеждат 
две стойности – например „уикенд“, ако денят е събота или неделя, и „работна 
седмица“ за останалите дни. За целта може да се използва следната формула (f7):

(f7)	 =IF(WEEKDAY(A4;2)>=6;“уикенд“; “работна седмица“)
Вторият аргумент ([return_type]) на функцията WEEKDAY указва как се но-

мерират дните от седмицата. Стойността на аргумента return_type в случая е 2, 
защото в България е прието седмицата и съответно номерирането на дните в нея 
да започва от понеделник. Аргументът return_typе не е задължителен и ако не 
се посочи изрично, Excel приема стойността му по подразбиране, която е 1, т.е. 
първият ден от седмицата ще бъде неделя, вторият – понеделник и т.н. След като 
е добавена подходяща колона (приемаме, че клетките в нея са именувани „Уик-
енд“), може лесно да се намери средната стойност на продажбите през работната 
седмица чрез следната формула (f8):

(f8)	 =AVERAGEIF(Уикенд;“работна седмица“;Стойност)
Тази формула е значително по-опростена и най-важното – връща коректен 

резултат, за разлика от f5. Много често, когато се решават по-сложни задачи, 
може да се окаже най-удачно да се добавят допълнителни колони. Разбира се не 
бива да се прекалява, защото прекомерното добавяне на колони прави таблицата 
тромава и трудна за работа.

2.2.2. Обобщаване на данни чрез пивот таблици (Pivot tables)

Пивот таблиците са мощно средство за обобщаване на данни от таблици в 
различни разрези. С тяхна помощ могат бързо и лесно да се оформят сложни от-
чети с различни нива на обобщаване, разрези и обобщаващи функции. Лекотата 
на работа с пивот таблиците и резултатите, които се постигат с тях, ги правят 
предпочитано средство за работа от страна на мениджъри, бизнес анализатори и 
широк кръг специалисти.

Възможностите им могат да се демонстрират с помощта на данните от фиг. 
2.1. С тяхна помощ могат да се получат същите резултати като показаните в таб-
лица 2.1. 

Преди да се създаде пивот таблица, трябва да са въведени подходящи дан-
ни. Не е задължително тези данни да са дефинирани като таблица (table)9, но е 
задължително да са оформени съгласно изискванията на Excel за таблица. Тези 
изисквания са:

•• имената на колоните да са разположени на първия ред, като не се допускат 
многоредови антетки10 на таблицата;

•• имената на колоните трябва да са уникални;
•• в колоните да са разположени еднотипни данни, например в колона „Дата“ 

да има само дати, а не текст, числа и др.;
•• в таблицата да няма празни редове или колони;
•• около таблицата е желателно да има поне по един празен ред и колона.

Тъй като таблицата от фиг. 2.1 отговаря на посочените изисквания, може 
да се премине към създаване на пивот таблицата. Колоните от таблицата, коя-
то се обобщава в пивот таблица, се възприемат като полета (Fields). Имената на 
полетата се вземат от съдържанието на клетките в първия ред на таблицата. На 
фиг. 2.2 е показана пивот таблица, която сумира продажбите по дати и градове с 
възможност за филтриране по дни от седмицата. На фигурата се виждат отдел-
ните области на една пивот таблица – Values (за полетата, чиито стойности се 
обобщават), Rows (за полетата, по чиито стойности се формират редовете в пивот 
таблицата), Columns (за полетата, по чиито стойности се формират колоните) и 
Filters (за полетата, по чиито стойности се филтрират данните в таблицата). Едно 
поле може да се разположи само веднъж в областите Rows, Columns или Filters, 
но няколко пъти в областта Values. Разполагането на едно поле повече от един път 
в областта Values има смисъл, ако се използват различни обобщаващи функции 
(SUM, MIN, MAX, AVERAGE и др.) или различни начини на представяне на об-
общените стойности (като процент от общата сума по редове, по колони, общо за 
таблицата и др.).
9	 Повече за таблици - в т. 2.3.1.
10	Антетка на таблица е първият ред от нея, който съдържа имената на колоните.

Фиг. 2.2. Структура на пивот таблица

Когато се разположи поле в областта Values, което съдържа числови данни, 
Excel автоматично прилага обобщаваща функция за сумиране (SUM). По жела-
ние може да се смени обобщаващата функция чрез избор на Summarize Values By 
или Value Fields Settings от конктекстното меню11 на клетка от областта Values. 
Възможните функции за обобщаване са: SUM, COUNT, MIN, MAX, AVERAGE, 
PRODUCT (произведение на стойностите), COUNT NUMBERS (брои само чис-
ловите стойности), както и някои статистически функции за стандартно отклоне-
ние и вариации. Полетата, които съдържат текстови данни, могат единствено да 
се обобщават с функция COUNT. 

Някои по-често използвани възможности на пивот таблиците са:
А) Групиране на данни по колони или редове
Групирането е полезен инструмент, когато полето в областта Rows или 

Columns има голям брой различни стойности и/или когато трябва да се изчислят 
междинни сборове за група от редове/колони. За целта първо се разполага в Rows 
или Columns полето, чиито стойности ще се групират, и след това се избира ко-
манда Group от контекстното меню на клетка от областта. Има някои особености 
на групирането в зависимост от типа на данните в полето. Ако полето съдържа 
дати, Excel дава възможност групирането да е по години, тримесечия, месеци, 
брой дни, часове, минути, секунди. Числовите полета могат да се групират на 
равни интервали, а текстовите – само чрез предварително селектиране на стойно-
стите, които ще се групират.

11	Контекстното меню на избран обект сe визуализира чрез натискане на десен бутон на мишката и 
съдържа най-често използваните команди за работа с обекта.

На фиг. 2.3 е показано групиране на дните от по
неделник до петък и тези през уикенда. След групира
нето се добавя ново поле, което може да се преиме-
нува и да се разполага в областите Rows, Columns и 
Filters на пивот таблицата. Стойностите на групите 
(Group 1 и Group 2) могат да се редактират, съответ-
но на „работна седмица“ и „уикенд“, и така да се 
получи пивот таблицата на фиг. 2.4, която сравнява 
среднодневни продажби по магазини през работната 
седмица и уикенда. В показаната таблица, след гру-
пиране на дните, Excel добавя ново поле – „Ден2“, 
чийто стойности съответстват на имената на двете 

групи – „работна седмица“ и „уикенд“. Полето „Ден2“ може по желание да бъде 
преименувано.

Фиг. 2.4. Пивот таблица с поле, получено при групиране

Б) Промяна на начина на представяне на резултатите
Много често се налага да се представят не само абсолютните обобщени 

стойности, но и относителните дялове, индексите, ранговете, натрупванията. 
Пивот таблиците позволяват визуализация на обобщените стойности в областта 
Values по различни начини. Най-често се представят относителните дялове чрез 
следните елементи от контекстното меню на клетка от областта Values и команда 
„Show Values as“: % of Grand Total (дял от общата стойност в пивот таблицата); 
% of Column Total (дял от общата стойност за колоната); % of Row Total (дял от 
общата стойност за реда). На фиг. 2.5 е представена пивот таблица, която показва 
различните начини на визуализация на относителните дялове за общата стойност 
на продажбите по магазини и работна седмица/уикенд. 

Фиг. 2.3. Групиране 
на текстови стойности

Column Labels
Row Labels Работна седмица Уикенд Grand Total

Варна
Sum of Стойност 71 209.00 лв. 28 964.00 лв. 100 173.00 лв.
% of Grand Total 23.60% 9.60% 33.20%
% of Column Total 32.38% 35.40% 33.20%
% of Row Total 71.09% 28.91% 100.00%

Русе
Sum of Стойност 70 535.00 лв. 26 026.00 лв. 96 561.00 лв.
% of Grand Total 23.38% 8.63% 32.00%
% of Column Total 32.07% 31.81% 32.00%
% of Row Total 73.05% 26.95% 100.00%

София
Sum of Стойност 78 179.00 лв. 26 820.00 лв. 104 999.00 лв.
% of Grand Total 25.91% 8.89% 34.80%
% of Column Total 35.55% 32.78% 34.80%
% of Row Total 74.46% 25.54% 100.00%

Total Sum of Стойност 219 923.00 лв. 81 810.00 лв. 301 733.00 лв.
Total % of Grand Total 72.89% 27.11% 100.00%
Total % of Column Total 100.00% 100.00% 100.00%
Total % of Row Total 72.89% 27.11% 100.00%

Фиг. 2.5. Пивот таблица с извеждане на относителни дялове

На фиг. 2.6 са показани три пивот таблици с други начини на визуализация 
на обобщените стойности. Обобщаващите функции (Summarize Values By) и на-
чинът на представяне (Show Values as) са дадени като коментари към клетките 
от областта Values. В посочените таблици в областта Values се разполага няколко 
пъти полето „Стойност“. Във втора и трета таблица полето „Дата“ в областта 
Rows е групирано по диапазон от 7 дни.

Фиг. 2.6. Пивот таблици 
с различни начини на представяне на обобщените стойности

В) Извличане на данни от пивот таблица
Обобщените данни в областта Values могат да се извличат от пивот таблица-

та чрез функция GETPIVOTDATA. Нейният формат е следният:
(f9)	 =GETPIVOTDATA(Data Field, Pivot table, Field1, Item1, Field2, 

Item2,….),
където Data Field е полето в пивот таблицата, чиято обобщена стойност връ-

ща функцията, Pivot table – абсолютен адрес на клетка от пивот таблицата, Field1 
– име на поле от областта Rows или Columns, Item1 – стойност на полето Field1. 
Функцията позволява да се зададат множество двойки Field и Item. На фиг. 2.7 

са показани различни варианти на аргументи на функцията GETPIVOTDATA за 
извличане на данни от пивот таблицата. Полето „Ден2“ е получено в резултат на 
групиране на стойностите на полето „Ден“.

Фиг. 2.7. Извличане на данни чрез GETPIVOTDATA

Задължителни аргументи на функцията GETPIVOTDATA са първите два 
– Data Field и Pivot table. Ако се пропуснат останалите аргументи, функцията 
GETPIVOTDATA връща обобщената за цялата таблица стойност (ред 11 на фиг. 
2.7), т.е. средната стойност на продажбите за целия период. На ред 8 е показан 
пример за извличане на средните продажби за Варна през работната седмица. 
Може да се сравни този начин за извеждане на усреднена стойност с използване 
на допълнителна колона и функция AVERAGEIF (f8). И двата варианта водят до 
един и същи резултат. В единия случай се налага добавяне на допълнителна коло-
на в основната таблица, а в другия – създаване на пивот таблица и извличане на 
данни от нея с GETPIVOTDATA.

Форматът на функция GETPIVOTDATA изглежда сложен, но Excel предлага 
автоматично въвеждане на функцията чрез въвеждане на знака „=“ и избор на 
клетка от областта VALUES или сборните редове/ко-
лони на пивот таблицата. За да може да се генерира 
GETPIVOTDATA със съответните аргументи обаче, 
трябва да е маркирана опцията Generate GetPivotData 
от меню Pivot Table Tools -> Analyze -> Pivot Table -> 
Options (вж. фиг. 2.8). Ако тази опция не е маркира-
на, то вместо GETPIVOTDATA ще се изпише адре-
сът на избраната клетка. Фиг. 2.8. Опция 

Generate GetPivotData

2.2.3.	 Задача за обобщаване на данни от таблица

За да се сравнят различните начини на обобщаване на данни, ще бъде пред-
ставена примерна задача и нейното решение. На фиг. 2.9 са показани условията, 
отнасящи се до оцветените клетки. 

Фиг. 2.9. Задача за обобщаване на данни

Решение
1. Изборът на една от няколко възможни стойности в Excel се реализира чрез 

инструмента Data Validation –  от меню Data. Чрез него могат да 
се наложат ограничения върху въвежданите в клетките данни. В случая ограни-

чението се отнася до списък с възможни стой-
ности, затова в диалоговия прозорец се избира 
Allow: List, а в полето Data Source се посочват 
стойностите, разделени със символа „ ; ”12 (вж. 
фиг. 2.10). Чрез опциите в списъка Allow могат 
да се приложат и други ограничения по отноше-
ние на въвежданите данни, като: Whole number 
(цели числа), Decimal (десетични числа), Date 
(дати), Time (часове), Text length (дължина на 
текстовия низ) и Custom (ограничение – фор-
мула). Чрез списъка Data се избира оператор за 

12	Символът, който разделя елементите в даден списък, се нарича list separator и се задава в регионал-
ните настройки на Windows. Обикновено е „ ; “ , но понякога може да е избран и символът „ ,“.

Фиг. 2.10. 
Data Validation

сравнение, чрез който се задава самото ограничение. Например, Allow: Whole 
number, Data: less than or equal to, Value: 2018, ще позволи въвеждане само на цели 
числа, по-малки или равни на 2018. Такова ограничение е подходящо за клетка, в 
която потребителите въвеждат година и се цели забрана за въвеждане на бъдеща 
година.

2. Това условие се решава по подобие на предишното с помощта на Data 
Validation.

3. В клетката A4 трябва да се извежда символен низ, който включва в себе си 
стойности на клетки. Може да се използва функцията CONCATENATE (f10) или 
операторът  &  за слепване на низове (f11): 

(f10)	 =CONCATENATE(„Общите продажби за “;B1;“ в “;B2;“ са: “)
(f11)	 =“Общите продажби за “ &B1 & “ в “ &B2& “ са:“
4. Решението е аналогично на условие 3.
5. Общите продажби за избрания град и ден от седмицата могат да се изчис-

лят по два начина: с използване на функция SUMIFS или чрез пивот таблица и 
GETPIVOTDATA. Формулата с използване на SUMIFS е следната:

(f12)	 =SUMIFS(Стойност;Ден;B2;Магазин;B1)
Ако се избере вариантът с пивот таблица за намиране на обобщените стой-

ности, може да се създаде една пивот таблица, която да се използва за пето и 
шесто условие на задачата. Пивот таблицата, подходяща за случая, е представена 
на фиг. 2.11.

Фиг. 2.11. Пивот таблица за решение на условия 5 и 6

За да се изведат в клетка В4 общите продажби за избрания в клетка В1 град 
и избрания в клетка В2 ден от седмицата, се използва следната формула (f13):

(f13)	 =GETPIVOTDATA(“Sum of Стойност“;$F$1;“Ден“;B2; 
“Магазин“;B1)

6. Средните продажби за избрания град и ден могат лесно да се извлекат от 
пивот таблицата от фиг. 2.11 с GETPIVOTDATA (f14):

(f14)	 =GETPIVOTDATA(“Average of Стойност“;$F$1; “Ден“;B2; 
“Магазин“;B1)

Формулата с използване на AVERAGEIFS изглежда по следния начин (f15):
(f15)	 =AVERAGEIFS(Стойност;Ден;B2;Магазин;B1)

2.3.	Свързване на таблици с помощта на функции

2.3.1.	 Работа с таблици

В Excel с термина „таблица“ (table, Excel table) се обозначава област от клет-
ки, оформена по определен начин и изрично дефинирана като такава. Получава се 
известно дублиране на понятията, тъй като обичайно „таблица“ се нарича област 
от клетки, организирана в колони и редове. Excel прави разграничаване между 
обикновена област от клетки (range) и таблица (table). Дефинирането на област от 
клетки като таблица дава много предимства и улеснява работата с данните. Excel 
предлага и полезни инструменти за работа с таблици, затова първо ще се разгле-
дат начините за дефиниране на таблица, както и това какви действия могат да се 
извършват върху нея. С цел избягване на двусмислието ще се използва понятието 
„таблица“ в смисъл на Excel table, а не област от клетки (range).

Не всяка област от клетки може да се зададе като таблица. Трябва да са спа-
зени редица изисквания, които бяха споменати в т. 2.2.2. 

Дефинирането на област от клетки като таблица може да стане по един от 
следните начини – чрез команда Format as table  и избор на желан стил на фор-
матиране (вж. фиг. 2.12) или чрез команда Insert -> Table . 

Когато се избере произволна клетка от таблица, става активно меню Table 
Tools . От него могат да се избират инструменти за работа с таблиците, 
като задаване на име на таблицата –    , обобщаване чрез пивот таблица 
– , премахване на дублиращи се редове – , преобразуване на таблицата в 
област от клетки13 – , добавяне на сборен ред14 – , добавяне/скриване 
на бутони за филтриране –  и др.

13	Това е обратното действие на дефиниране на таблица.
14	Обобщаващата функция на сборния ред може да се избира измежду обобщаващите функции, 

използвани в пивот таблиците. Може да се посочва кои колони да се обобщават в сборния ред.

Фиг. 2.12. Дефиниране на таблица

Когато се работи с таблица, има някои особености при адресирането на клет-
ките в нея (вж. фиг. 2.12). Excel автоматично създава име за цялата таблица, както 
и имена на всяка от колоните в нея. Имената на новите таблици се задават от 
Excel като Table1, Table2 и т.н., но потребителят може по желание да редактира 
името по подразбиране. Така вместо адрес на таблицата (А5:D9) се използва ней-
ното наименование (Стоки)15, a вместо адрес на клетки от колона (напр. А5:А9) 
– Стоки[Код]). Трябва да се има предвид, че името на таблица (напр. „Стоки“) е 
равносилно на използване на абсолютния адрес на клетките в нея ($А$5:$D$9). 
Относителните адреси на клетки от една колона в таблицата се заместват от „@
[име на колоната]“ във формула в друга колона (вж. формулата в колона „Цена с 
ДДС“ на фиг. 2.12).

При свързване на две или повече таблици с помощта на вградени функции 
е препоръчително да се работи с таблици, а не с области от клетки. Когато се ра-
боти с дефинирани таблици, формулите стават по-лесни за възприемане – вместо 
адреси като аргументи се използват имена на таблици и колони, които са по-раз-
бираеми за потребителите.

2.3.2.	 Функции за търсене и обръщение (Lookup and Reference)

В тази категория спадат функции, чрез които могат да се извличат данни 
от една таблица в друга. Най-често данните, които се извличат, са относително 
постоянни (т.нар. номенклатури). Tакива са списъците с клиенти, стоки, цени и 
15	Името на таблицата е зададено чрез инструмента Table Name от меню Table Tools.

т.н. Чрез подходящи функции е възможно постоянните данни да се въвеждат ед-
нократно и да се използват многократно в различни таблици. Друг често срещан 
сценарий за работа с функции за търсене и обръщение е когато се търси в диа-
пазон, например ако трябва да се изчисли отстъпка в зависимост от стойността 
на продажбата. Използването на функция IF, когато има повече от три-четири 
диапазона, е доста усложнено и поради това по-удачният начин е чрез използване 
на VLOOKUP, LOOKUP и HLOOKUP.  

VLOOKUP, HLOOKUP, LOOKUP
Форматът на функцията VLOOKUP е:
(f16)	 VLOOKUP(lookup_value; table_array; col_index_num [;range_ 

lookup])
VLOOKUP се използва при „вертикално“ търсене, т.е. когато претърсвана-

та област table_array е организирана по колони, какъвто е случаят с таблиците. 
Функцията VLOOKUP търси lookup_value в най-лявата колона на претърсвана 
област table_array и връща резултат от намерените ред и колона, с пореден номер 
в претърсваната област col_index_num. Аргументът range_lookup определя начи-
на на търсене и има две възможни стойности:

•• FALSE – за търсене по точна стойност. Търси се точно съвпадение на 
lookup_value и стойностите в най-лявата колона на областта table_array. Ако не 
намери точно съвпадение, функцията връща грешка от тип #N/A! (not available). 
При този начин на търсене не се изисква областта table_array да е сортирана по 
нарастващи значения на стойностите в най-лявата колона;

•• TRUЕ – за търсене в диапазон. Когато не се намери търсената стойност, 
функцията връща резултат от ред, чиято стойност в най-лявата колона на област
та table_array е най-близка, но ненадвишаваща търсената. За да върне коректен 
резултат, се изисква областта table_array да бъде сортирана по нарастващи значе-
ния на клетките в най-лявата колона. По подразбиране аргументът range_lookup 
има стойност TRUE.

На фиг. 2.13 е показан пример с използване на функция VLOOKUP за търсе-
не по точна стойност. В клетка С11 е въведен код на стока, а в клетки С12 и С13 се 
извличат с VLOOKUP съответно наименованието и цената без ДДС на стоката. В 
този случай е задължително да се укаже, че търсенето е по точно съвпадение, като 
се въведе стойност FALSE (или 0) за последния аргумент на функцията range_
lookup. Ако не се посочи стойност FALSE, се приема стойността по подразбиране 
(TRUE) и функцията VLOOKUP може да върне некоректни резултати. Например 
ако в C11 се въведе код „104“, VLOOKUP с аргумент FALSE в клетка C12 ще се 
върне грешка #N/A!, а VLOOKUP с аргумент TRUE – наименованието на стоката 
с код „103“, защото 103 е най-близка до търсената, без да я надвишава. 

Фиг. 2.13. Търсене по точна стойност с VLOOKUP

Полезна практика е да се обработват грешките чрез подходящи функции 
като IFERROR, IFNA, ISERR, ISERROR. В конкретния пример, за да се посочи на 
потребителя, че е въвел грешен код на стока, може да се модифицира формулата 
в клетка С12 по следния начин:

(f17)	 =IFNA(VLOOKUP(C11;Стоки;2;FALSE);“Грешен код“)
Първият аргумент на функцията IFNA e VLOOKUP. Ако VLOOKUP намери 

кода от клетка C11, формулата (f17) ще върне съответната стойност. Ако обаче 
липсва съответствие, ще се изпише текстът „Грешен код“. Функцията IFNA про-
верява само за грешка от тип #N/A!, т.е. когато не е намерен въведеният в клетка 
C11 код. Може да се използва и функция IFERROR, но няма сигурност, че кодът е 
грешен, защото IFERROR проверява за всички типове грешки, а не само за #N/A!. 

Когато претърсваната област е организирана по редове, се използва функ-
ция HLOOKUP. На фиг. 2.14 е показан пример за използване на VLOOKUP 
и HLOOKUP за търсене в диапазон, което се указва с последния аргумент на 
функциите (range_lookup = TRUE). В областта от клетки „Оценки“ се изчислява 
оценката на всеки студент по шестобалната система в зависимост от общия брой 
точки, съгласно правилата за оценяване в Икономически университет – Варна. 
Претърсваните области (A4:B8 и E3:I4), както и данните за оценките на студен-
тите (D9:H14), не са дефинирани като таблици, затова във формулите участват 
адресите на клетките. Примерът може да се реши и с дефиниране на таблици, но 
следва да се има предвид, че това е възможно само за областите от клетки, в които 

данните са организирани по колони, а не по редове. Това означава, че областта 
D3:I4 няма смисъл да се дефинира като таблица, тъй като тя не отговаря на изис-
кванията за такава.

Фиг. 2.14. Търсене в диапазон с VLOOKUP и HLOOKUP

За да могат функциите VLOOKUP и HLOOKUP да върнат коректен резул-
тат при търсене в диапазон, е задължително данните в най-лявата колона (при 
VLOOKUP) и първия ред (при HLOOKUP) на претърсваната област трябва да са 
сортирани по възходящ ред.

Аргументът range_lookup при търсене в диапазон задължително трябва да е 
TRUE или да не се посочва. Ако range_lookup е FALSE, функциите HLOOKUP и 
VLOOKUP в примера на фиг. 2.14 ще върнат грешки #N/A!. 

Функцията LOOKUP също може да се използва за търсене в диапазон. Тя 
има два формата, като по-често използваният е:

(f18)	 LOOKUP (lookup_value; lookup_vector [; result_vector)
 Функцията търси lookup_value във вектора lookup_vector и връща стойност-

та, която се намира на същата относителна позиция в result_vector. Lookup_vector 
и result_vector са области от клетки, съдържащи един ред или една колона. Изис-



48	 Ръководство за подготовка за национално състезание по информатика

ква се lookup_vector и result_vector да са с едни и същи размери, т.е. да са редове/
колони с еднакъв брой клетки. LOOKUP може да се използва, когато се търси в 
произволна колона/ред, а не в най-лявата колона, както е при VLOOKUP или в 
първия ред, както е при HLOOKUP. За да върне коректни резултати, LOOKUP из-
исква стойностите в lookup_vector да са сортирани във възходящ ред. Функцията 
LOOKUP не може да се използва при търсене по точна стойност, освен ако не е 
спазено горното условие. Препоръчва се вместо LOOKUP да се използват функ-
циите HLOOKUP и VLOOKUP. 

INDEX, MATCH
Функциите VLOOKUP и HLOOKUP са много подходящи при свързване на 

таблици. Te обаче имат едно сериозно ограничение – VLOOKUP търси винаги 
в най-лявата колона, а HLOOKUP – само в първия ред от претърсваната област. 
LOOKUP може да търси в произволен ред, но само в сортиран вектор, и то само 
в диапазон. Понякога може да се наложи търсене по точна стойност в други ко-
лони и редове от претърсваната област. В тези случаи се използва комбинация от 
функциите INDEX и MATCH.

Форматът на функция INDEX е:
(f19)	 =INDEX (array; row_num [; column_num])
Резултатът, който връща функцията, е стойността на клетката, намираща се 

на указания ред (row_num) и колона (column_num) в претърсваната област array. 
За намиране на ред и колона в областта array се използва функция MATCH. Ней-
ният формат е следния:

(f20)	 =MATCH (lookup_value; lookup_array; [match_type])
Функцията търси lookup_value в областта lookup_array и връща относител-

ната позиция на намерения резултат в рамките на lookup_array. Тук, за разлика от 
VLOOKUP и HLOOKUP, областта, в която се търси, трябва да е колона или ред, 
а не таблица с няколко колони и/или редове. Аргументът match_type указва какво 
съответствие се търси и има три възможни стойности:

•• 1 – търси се стойност, по-малка или равна на lookup_value. Това е стой-
ността на аргумента по подразбиране;

•• 0 – търси се точно съответствие на lookup_value;
•• –1 – търси се стойност, по-голяма или равна на lookup_value.

На фиг. 2.15 е показан пример за използване на функции INDEX и MATCH. 
В областта „Валутни курсове“ (А3:F8) са въведени кръстосани курсове за основ-
ни валути. Задачата е да се създаде валутен калкулатор, който да изчислява какво 
количество от дадена валута може да се купи при въведена валута в клетка I4 и 
количество в клетка J4. За клетки I4 и I5 е добре да се зададе ограничение за въ-
вежданите данни (Data Validation) от тип списък, който да включва всички кодове 
на валути (напр. A4:A8). Валутата, от която ще се преобразува, се въвежда/избира 

в клетка I4, а валутата, към която се изчислява, се въвежда/избира в клетка I5. В 
клетка J4 се въвежда количество от първата валута, а в клетка J5 с формула трябва 
да се изчисли количество от втората валута, което може да се закупи с количест
вото от първата валута.

Фиг. 2.15. Търсене с INDEX и MATCH

За формулата в клетка J5 има два основни варианта. Първият е с използ-
ване на комбинация от функциите INDEX и MATCH, както е показано на 
фиг. 2.15, а вторият вариант – с използване на функциите VLOOKUP и MATCH. 
При първия вариант като първи аргумент на INDEX се задава областта от клетки 
с данни за валутните курсове. Вторият аргумент е резултатът, който връща функ-
ция MATCH, търсеща относителната позиция на валутата в I4 сред валутите в 
областта A4:A8. Валутата EUR се намира на втора позиция. Аналогично третият 
аргумент на INDEX, функцията MATCH, определя номера на колоната, от която 
ще се върне резултат – 3. Като резултат INDEX връща стойността на клетката, на-
мираща се на втори ред и трета колона в областта B4:F8. В случая това е клетката 
с адрес D5. 

Тъй като валутата, от която ще се преобразува, се намира в най-лявата ко-
лона на областта „Валутни курсове“, може да се използва и функция VLOOKUP 
в комбинация с MATCH. В този случай формулата в клетка J5 от фиг. 2.15 ще 
изглежда по следния начин (f21):

(f21)	 =VLOOKUP(I4;A4:F8;MATCH(I5;A3:F3;0); FALSE)*J4
Функцията MATCH се използва за изчисляване на колоната в претърсваната 

област (A4:F8), от която трябва да се върне резултат.
За да се оцветява избраният валутен курс, е приложено условно форматира-

не (Conditional Formatting) на клетките в областта B4:F8 (вж. фиг. 2.16). Прави-
лото за условно форматирано е зададено с формула с използване на функцията 
AND. Функцията AND е логическа и връща резултат истина (TRUE), ако всички 
нейни аргументи – условия, са верни. В примера условията са клетката в колона 
А (зададена чрез смесен адрес $A4) да е равна на въведената валута в I4 и клетка-
та от ред 3 (смесен адрес B$3) да е равна на валутата в I5. Важно е да се използват 
смесените адреси $А4 и B$3. Чрез тях се указва, че за всички клетки от маркира-
ната област (B4:F8) колона А и ред 3 са фиксирани. 

Фиг. 2.16. Условно форматиране с използване на формула

Задачи от тип „класиране“ с INDEX, MATCH, SMALL и LARGE
Функциите INDEX и MATCH могат да се използват за решаване на зада-

чи от типа „Изведете трите най-продавани продукта“, „Кои са петте клиенти с 
най-ниски стойности на продажбите?“ и т.н., т.е. когато се решават задачи от тип 
„класиране“. Ако трябва да се изведат само най-високите или най-ниските стой-
ности, могат да се използват функциите SMALL и LARGE. Форматът на функци-
ята LARGE е:

(f22)	 =LARGE (array, К)
Функцията връща К-тата поред стойност по големина от областта – масив 

array. Аналогичен е форматът и на функцията SMALL. За да се изведе съдържа-
нието на клетки, които съответстват на тези стойности, трябва да се използват 
функциите INDEX и MATCH. 

На фиг. 2.17 е показан пример за прилагане на функциите SMALL, LARGE, 
INDEX и MATCH при решаване на задачи от тип „класиране в нарастващ или 
намаляващ ред“. В областта A4:D15 са въведени данни за стойностите на три 
показателя за 11 държави. В табличната област F3:J7 е направено класиране по 
най-високи и по най-ниски стойности на населението на държавите.

Фиг. 2.17. Класиране с INDEX, MATCH, SMALL и LARGE

Трите най-високи (H5:H7) и трите най-ниски стойности (J5:J7) се изчисляват 
с помощта на функции LARGE и SMALL. За извеждане на имената на държави-
те, които имат тези стойности, се използва комбинацията от INDEX и MATCH. 
Принципът на работа е следният: търси се относителната позиция на k-тата спо-
ред стойност в колоната „Население (млн)“ (MATCH) и се връща клетката, коя-
то се намира на същата относителна позиция в колона „Държава“ (INDEX). Тук 
трябва да се отбележи едно ограничение на този начин на класиране. Ако има ня-
колко държави с едни и същи търсени стойности на население, функцията INDEX 
ще върне името на първата намерена. В такива случаи могат да се използват дру-

ги средства за класиране – например условно форматиране на табличната област 
(Top/Bottom Rules) и след това филтриране по цвят (Filter By Color). Ако трябва 
да се извлекат данни за редовете с максимални/минимални стойности в отделна 
таблица, може да се използва и разширено филтриране (Advanced Filter). 

Задача за самостоятелна работа
Примерът от фиг. 2.17 се усложнява, като целта е да се направи класиране 

по избран показател, а не само по население. За клетка H1 трябва да се зада-
де такова ограничение, че да може да се избира един от трите показателя (вж 
фиг. 2.18). Формулите в областта G5:J7 трябва да се променят така, че да се из-
веждат най-високите/най-ниските стойности по избрания показател, както и съ-
ответните държави. 

Фиг. 2.18. Класиране по избран показател (задача за самостоятелна работа)

Указания по задачата: Функцията INDEX има два формата. Единият е този, 
който вече беше представен: формулата връща съдържание на клетка. Други-
ят формат на функцията обаче позволява да се връща не съдържание на клетка, 
а адрес на клетка/област от клетки. При въвеждане на функция INDEX може да 
се избира с кой от двата формата да се работи. Когато аргументът row_num или 
column_num е 0, функцията INDEX връща адрес на цяла колона или цял ред. На-
пример ако с функция INDEX се цели да се върне адресът на клетките от колона 
„Продължителност на живота (жени)“ от примера на фиг. 2.17, трябва да се из-
ползва функцията INDEX със следните аргументи: INDEX($B$5:$D$15;0;2), т.е. 
адресите на клетките от колона, чиято относителна позиция в областта B5:D15 е 2.

Тази самостоятелна задача е с по-голяма трудност от задачите за състезание 

и е предизвикателство за тези, които искат да изпробват нови начини за решаване 
на такива задачи и да научат повече за възможностите на Excel.

2.4.	Задания за подготовка
В тази точка е представено решението на две задания – комплексно зада-

ние с използване на разгледаните функции и задание от състезанието в секция 
„Microsoft Excel“ за 2018 г. В края на точката е дадено задание за подготовка за 
състезанието.

2.4.1. Комплексно задание 

Обменно бюро „Варна Трейд“ търгува с основни валути. В таблица „Кур-
сове“ (вж. фиг. 2.19) са въведени курсовете за купуване и продаване на валутите, 
както и централният курс на БНБ. 

Данните за покупките и продажбите на валута се въвеждат в таблица „Бор-
дера“ (вж. фиг. 2.20). Всяко бордеро отразява една операция по покупка или про-
дажба. Покупката и продажбата се разглеждат от страната на обменното бюро, 
т.е. „купува“, когато обменното бюро купува валута от клиент, а „продава“ – кога-
то бюрото продава на клиент.

Фиг. 2.19. Таблица „Курсове“ Фиг. 2.20. Таблица „Бордера“

В таблица „Отчет“ (вж. фиг. 2.21) се прави рекапитулация на операциите за 
деня. Наличността в началото на деня се въвежда за всяка валута. Изчисляват се 
стойността на наличността в лв. по курса на БНБ, приход, разход и наличност в 
края на деня в съответната валута и в лв. по курс на БНБ.

Фиг. 2.21. Таблица „Отчет“

Задачи за работа:
1.	В колона „Стойност в лв.“ от таблица „Бордера“  (фиг. 2.20) да се въведе 

формула за изчисляване на стойността в лв. като произведение на количество и 
курса за дадената валута и вид операция („купува“ или „продава“) от таблица 
„Курсове“ (фиг. 2.19).

2.	В колони „Наличност в началото на деня в лв. по курс на БНБ“ и „Налич-
ност в края на деня в лв. по курс на БНБ“ от таблица „Отчет“ (фиг. 2.21) да се 
изчисли стойността като произведение на количество и курса на БНБ за дадената 
валута от таблица „Курсове“.

3.	В колона „Приход“ от таблица „Отчет“ (фиг. 2.21) да се сумират постъпи-
лите количества от операции с дадената валута. Приходът се изчислява по след-
ния начин: ако валутата в колона F е BGN – се сумира стойността в лв. от таблица 
„Бордера“ за операциите от вид „продава“; за останалите валути се сумира коли-
чеството от таблица „Бордера“ за съответната валута и вид операция „купува“. 

4.	В колона „Разход“ от таблица „Отчет“ се изчислява разходът по валути по 
следния начин: ако валутата е BGN, се сумира стойността в лв. от таблица „Бор-
дера“ за вид операция „купува“; за останалите валути се сумира количество от 
таблица „Бордера“ за дадената валута и вид операция „продава“.

5.	В колона „Наличност в края на деня“ се изчислява оставащото количест
во в различни валути по следната формула: [Наличност в началото на деня] + 
[Приход]-[Разход].

Решение
Решението приема, че трите области – „Курсове“, „Бордера“ и „Отчет“ – са 

дефинирани като таблици със същите имена. Приемат се като равносилни и фор-
мули без използване на таблици.

6.	Формулата в клетка Е13 може да се въведе по различни начини. В нея тряб-
ва да се съобрази видът на операцията – ако е „купува“, то функцията VLOOKUP 
трябва да върне резултат от втората колона, ако е „продава“ – от третата колона 
на таблица „Курсове“. Номерът на колоната може да се изчисли или с използване 
на функция IF (f23) или с MATCH (f24). Тъй като има само два варианта, по-оп-
ростена би била формулата с IF:

(f23)	 =VLOOKUP([@Валута];Курсове;IF([@[Вид операция]]=“Купува“ 
;2;3);FALSE)* [@Количество]

 (f24)	 =VLOOKUP([@Валута];Курсове;MATCH([@[Вид операция]]; Курс
ове[[#Headers];[Купува]:[Продава]];0)+1;FALSE)* [@Количество]

Така показаните формули не проверяват за грешно въведени вид на опера-
ция и код на валута. За такава проверка могат да се използват функции IFNA или 
IFERROR и да се изведе подходящ текст от вида: „Грешна операция“ или „Греш-
на валута“. Друг вариант е да се наложат ограничения върху въвежданите данни 
чрез Data Validation. Ограниченията и за двете колони („Вид операция“ и „Валу-
ти“) трябва да са от тип списък (list), а в полето Source да се зададат възможните 
стойности: за вид на операция – „купува“ и „продава“; за валута – адресът на 
областта от клетки с кодовете на валутата от таблица „Курсове“ – А4:А8. 

7.	Формулата в H4 в таблица „Отчет“ е:
(f25)	 =VLOOKUP ([@Валута];Курсове;4;FALSE)* [@[Наличност в 

началото на деня]] ,
а на тази в L4:
(f26)	 =VLOOKUP ([@Валута];Курсове;4;FALSE)* [@[Наличност в 

края на деня]]
8.	В клетката I4 от колона „Приход“ се въвежда следната формула:
(f27)	 =IF([@Валута]=“BGN“;SUMIF(Бордера[Вид операция]; 

“Продава“;Бордера[Стойност в лв.]); SUMIFS(Бордера[Количество]; 
Бордера[Валута];[@Валута];Бордера[Вид операция];“Купува“)) 

Оценява се и решение, макар и с по-малко точки, без използване на IF, a с две 
различни формули – за BGN и за останалите валути.

9.	В клетка J4 се въвежда следната формула:
(f28)	 =IF([@Валута]=“BGN“; SUMIF(Бордера[Вид опера-

ция]; “Купува“;Бордера[Стойност в лв.]);SUMIFS(Бордера [Количество]; 
Бордера[Валута];[@Валута];Бордера[Вид операция];“Продава“))

10.	Формулата за изчисляване на наличност в края на деня в K4 е:
(f29)	 =[@[Наличност в началото на деня]]+[@Приход]–[@Разход]
В таблица 2.2 са показани същите формули, но при условие че не са дефини-

рани таблиците „Курсове“, „Отчет“ и „Бордера“. Формулите се различават един-
ствено по начин на адресиране на клетките. Когато не са дефинирани таблици, 
като аргументи на функциите участват адресите на клетките, а при дефинирани 

таблици – техните имена. И двата варианта на решения се приемат и оценяват 
еднакво.

Таблица 2.2
Формули за решаване на задачите с използване на адреси вместо имена на клетки

№ Формула 

f23 =VLOOKUP(C13;$A$4:$D$8;IF($B13="Купува";2;3);FALSE)*D13 

f24 =VLOOKUP(C13;$A$4:$D$8;MATCH($B13;$B$3:$C$3;0)+1;FALSE)*D13

f25  =VLOOKUP(F4;$A$4:$D$8;4;FALSE)*G4

f26 =VLOOKUP(F4;$A$4:$D$8;4;FALSE)*K4

f27 =IF(F4="BGN";SUMIF($B$13:$B$19;"Продава";$E$13:$E$19);SUMIFS($D$13:$D$19
;$C$13:$C$19;F4;$B$13:$B$19;"Купува")) 

f28 =IF(F4="BGN"; SUMIF($B$13:$B$19;"Купува";$E$13:$E$19); SUMIFS($D$13:$D$19;
$C$13:$C$19;F4;$B$13:$B$19;"Продава"))

f29 =G4+I4-J4

Допълнителна задача за самостоятелна работа: Да се изчислят прихо-
дът и разходът на валутите в таблица „Отчет“, като се използват пивот таблици 
и функцията GETPIVOTDATA. За да може GETPIVOTDATA да връща необходи-
мия резултат, е необходимо да се обмисли добре структурата на пивот таблицата.

2.4.2.	 Задание от състезанието през 2018 г.

В лист „Стоки“ (A3:D23) се намира списък с продаваните стоки (вж. фиг. 
2.22). Всяка от тях има уникален код, име, мярка и принадлежи на една група. 
Групите са общо 4 – „Плодове и зеленчуци“, „Мляко и млечни изделия“, „Месо и 
месни продукти“ и „Безалкохолни напитки и бира“.

В лист „Продажби“ (A3:K332) се намира таблица с данни за дневните про-
дажби на стоките за периода 18/12/2017 – 16/01/2018 г. (вж. фиг. 2.22). На всеки 
ред от таблицата се въвежда информация за дневните продажби на всяка стока за 
съответната дата. Някои от стоките са в промоция на определени дати. За стоките 
в промоция е въведен процент на отстъпка, който се прилага спрямо регулярната 
цена. Ако стоката не е в промоция, тя се продава на регулярна цена.

Фиг. 2.22. Лист „Стоки“ и лист „Продажби“

Задача 1. В оцветените клетки в лист „Продажби“ да се въведат подходящи 
формули, съгласно коментарите към заглавията на колоните:

1.	Колона „Ден“ – да се изведе името (или номерът) на деня от седмицата. 
Приемат се имена на английски (Monday, Mon), български (понеделник, пон) или 
пореден номер на деня от седмицата (1,2,...7).

2.	Колона „Име на стока“ – да се изведе с подходяща функция името, което 
съответства на въведения код от колона С, съгласно таблицата в лист „Стоки“.

3.	Колона „Група стоки“ – да се изведе с подходяща функция групата сто-
ки, която съответства на въведения код от колона С, съгласно таблицата в лист 
„Сток“.

4.	Колона „Цена с отстъпка“ = [Регулярна цена] – [Регулярна цена] * [От-
стъпка %].

5.	Колона „Стока в промоция“ – aко има отстъпка, да се изписва „да“, в про-
тивен случай – „не“.

6.	Колона „Стойност“ = [Цена с отстъпка] * [Количество].
Решение
Формулите в първите клетки в съответните колони са дадени в таблица 2.3. 

Показаните решения са без дефиниране на таблици.

Таблица 2.3
Формули за решаване на задача 1

Клетка Формула

B4 =TEXT(A4;"dddd") или =WEEKDAY(A4;2)

D4 =VLOOKUP(C4;Стоки!$A$4:$D$23;2;FALSE)

E4 =VLOOKUP(C4;Стоки!$A$3:$D$23;4;FALSE)

Клетка Формула

I4 =G4-G4*H4

J4 =IF(H4>0;"да";"не") или =IF(ISBLANK(H4);"не";"да") или др.

K4 =I4*F4

Задача 2. В лист „Обобщена“ да се изчисли общата стойност на продажбите 
по групи за целия период и относителният дял на продажбите за всяка група. 

Решение
Задачата може да се реши с обобщаващи функции или с пивот таблици (вж. 

фиг. 2.23).

Фиг. 2.23. Два варианта за решение на задача 2

Задача 3. Въз основа на получените стойности от задача 2 в лист „Обобще-
на“ да се създаде кръгова графика (Pie), представяща относителните дялове на 
продажбите по групи. Графиката да има заглавие, легенда и етикети (Data Labels).

Решение
Примерна графика, решение на задачата, 

е показана на фиг. 2.24. Приема се за вярно ре-
шение и пивот графика (Pivot Chart).

Задача 4. В лист Pivot да се създаде обоб-
щена таблица, представяща средната стойност 
на продажбите (колона „Стойност“) по дни от 
седмицата и групи стоки. Клетките в таблица-
та да се форматират с условно форматиране 

Фиг. 2.24. Решение на задача 3

(Conditional Formatting) по такъв начин, че да се оцветява най-голямата средна 
стойност за всяка група. 

Решение
Пивот таблицата е показана на фиг. 2.25.

Фиг. 2.25. Средна стойност по дни и групи стоки

Условното форматиране на средните стойности може да бъде направено по 
различни начини. В ръководството се предлагат два от тях:

•• Последователно прилагане на условно форматиране на клетките от облас
ти B3:B9, C3:C9, D3:D9 и E3:E9. Правилото за условното форматиране е „Top 1“;

•• Прилагане на условно форматиране за всички клетки в областта B3:E9. 
Правилото е от тип формула (use a formula to determine which cells to format) и се 
задава така:

(f30)	 =B3=MAX(B$3:B$9)
Това правило изисква да се маркират всички клетки от областта B3:E9, пре-

ди да се избере Conditional Formatting. Аргументът на функцията MAX трябва да 
е зададен чрез смесен адрес на клетките от областта B$3:B$9, a адресът на пър-
вата клетка от маркираната област B3 е нужно да е относителен. При прилагане 
на правила за условно форматиране от тип „формула“ формулата се изчислява за 
всяка клетка от областта и ако условието е изпълнено, се прилага зададеният фор-
мат. Относителният адрес на B3 позволява да се замести с адреса на всяка клетка 
в областта. Колоните в адреса B$3:B$9 са относителни, за да може при прилага-
нето на правилото всяка стойност да се сравни с максималната в колоната. Напри-
мер при прилагане на правилото за клетка C7 то ще изглежда по следния начин: 
=C7=MAX(C$3:C$9).

Задача 5. Към обобщената таблица от задача 4 да се създаде подходяща гра-
фика.

Решение
Създава се пивот графика (Pivot Chart). Подходящи са типове Clustered Bar, 

Clustered Column или Line. Графика тип Line се прилага само ако по оста X са 
разположени дните от седмицата. 

Допълнителна задача. С подходящи средства в лист „Допълнителна зада-
ча“ изчислете среднодневните продажби за всяка стока по време на промоция 
и без промоция. Има ли стоки, при които по време на промоция се реализират 
по-ниски стойности на продажбите в сравнение със среднодневните продажби на 
същата стока в дните, когато стоката не е на промоция? Кои са тези стоки? Оцве-
тете ги с условно форматиране.

Решение
Допълнителната задача носи бонус точки. Тя може да се реши по два начи-

на – с използване на функция за обобщаване (AVERAGEIFS) и с използване на 
пивот таблици (вж. фиг. 2.26).

Фиг. 2.26. Два варианта за решаване на допълнителната задача

Условното форматиране се задава с правило от тип формула. За да се оцветят 
целите редове със стоките, за които продажбите по време на промоция са по-ни-
ски от продажбите без промоция, трябва да се маркират всички клетки в областта 
A5:D24, респективно F5:H24. Формулата за условното форматиране е:

(f31)	 =$C5<$D5,
за клетките в областта A5:D24 или:
(f32)	 =$G5<$H5,
за клетките в областта F5:H24.

2.4.3.	 Задание за самостоятелна работа

Фирма „Паркет“ ООД продава масивни паркети. В лист „Номенклатури“ 
(вж. фиг. 2.27) са въведени данни за клиентите на фирмата (А3:B8), продавачите 
(А12:D17) и цените на различните видове паркет за 1  кв.м. (D3:H7). Паркетът 
може да бъде различно качество, което се отразява на неговата цена за кв.м.

Фиг. 2.27. Лист „Номенклатури“

В лист „Продажби“ (вж. фиг. 2.28) се въвеждат данни за продажбите на пар-
кети за месец октомври 2018 г. Всяка продажба има пореден номер, с една про-
дажба се продава един паркет на един клиент от един продавач.  

Фиг. 2.28. Таблица „Продажби за месец октомври 2018 г.“

Задача 1. В оцветените клетки от лист „Продажби“ да се въведат подходящи 
формули, така че:

1.	„Име на продавач“, „Име на клиент“ и „Име на паркет“ да се извличат с 
подходящи функции от таблиците от лист „Номенклатури“.

2.	Единичната цена на паркета да се извлича от таблица „Цени на паркети за 
кв.м.“ в зависимост от кода на паркета и неговото качество.

3.	Стойността е произведение на количество и единична цена.
Задача 2. В таблица „Продавачи“ от лист „Номенклатури“ да се сумира 

стойността на продажбите от лист „Продажби“ за всеки от продавачите в коло-
на „Оборот“.  В колона „Бонус“ с подходяща формула да се изчисли бонусът за 
всеки продавач в зависимост от оборота по следната схема: за оборот от 5000 до 
10 000 лв. – 200 лв.; над 10 000 до 15 000 лв. – 280 лв.; над 15 000 до 20 000 лв. – 
420 лв.; над 20 000 до 25 000 лв. – 600 лв. и над 25 000 лв. – 800 лв. За оборот до 
5 000 лв. не се полага бонус.

Задача 3. На нов лист да се създаде пивот таблица, която да представя об-
щото количество по име на паркет и качество. С условно форматиране да се оцве-
тят трите най-големи стойности на количество. Да се създаде подходяща графика 
към пивот таблицата.

Задача 4. С подходящи средства да се представят брой продажби и обща 
стойност на продажбите по седмици (1.10 – 7.10, 8.10 – 14.10, 15.10 – 21.10 и т.н.).

Допълнителна задача. Да се сумират броят и стойностите на продажбите 
по клиенти. В клетка извън таблицата с формула да се изведе името на клиента с 
максимална стойност на продажбите. 

